
Mehrfachauswahl im Datenüberprüfung-Dropdown 

 

Schritt 1: 

Rückkopplung der Dateneingabe in die Datenliste: 

Quelle der Länder: Tabellenblatt: Definition / Spalte A 

Listenbereich für Datenüberprüfung: Tabellenblatt: Quelle 1 Spalte / A 

Formel für Rückkopplung: 

=" - "&TEXTVERKETTEN(ZEICHEN(10);WAHR;Definition!A2;'Ausgabe 1'!B$2) 

- Aufzählungszeichen und Zeilenumbruch nach jedem Land 

- Liste der Länder 

- Aktuelle Auswahl 

 

Schritt 2: 

Verhindern von Mehrfachnennung einzelner Länder 

Rückkopplung der Auswahl nur noch, wenn die das Land der betreffenden Zelle nicht bereits in der 

Auswahl enthalten ist, also die Suche danach einen Fehler liefert: 

=WENN(ISTFEHLER(FINDEN(Definition!A2;'Ausgabe 2'!B$2));" - "& TEXTVERKETTEN(ZEICHEN(10); 

WAHR;Definition!A2;'Ausgabe 2'!B$2);"") 

- Überprüfung ob Land schon in der Auswahl enthalten ist 

- Aufzählungszeichen und Zeilenumbruch nach jedem Land 

- Liste der Länder 

- Aktuelle Auswahl 

 

Schritt 3: 

Verhindern von möglichen Lücken im Dropdown 

Die Im Schritt 2 erstellte Liste wird durch die Filterfunktion bereinigt und Leerzeilen gelöscht 

=FILTER(A1:A20;(A1:A20<>"")) 

Jetzt muss berücksichtigt werden, dass in der Datenüberprüfung auf die Zelle verwiesen werden 

muss, in der sich die Filterfunktion befindet. Da dies nun ein area ist, muss # an den Verweis angefügt 

werden: Also ='Quelle 3'!$A$24# 

 

 



Schritt 4: 

Erweiterung der Liste auf mehr Mitarbeiter. 

Um die Liste auf mehr Mitarbeiter zu erweitern müssen  die Zellenverweise dynamisch gestaltet 

werden. Jede Zeile in dem Ausgabe Tabellenblatt benötigt seine eigene Spalte in der Quelle als Liste. 

4.1 Dynamisch machen der Quelle 

=WENN(ISTFEHLER(FINDEN(Definition!A2;'Ausgabe 2'!B$2));" - "& TEXTVERKETTEN(ZEICHEN(10); 

WAHR;Definition!A2;'Ausgabe 2'!B$2);"") 

Der markierte Teil in der Rückkopplung muss dynamisch gestaltet werden, damit jede Spalte in der 

Quelle mit einer Zeile in der Ausgabe rückkoppelt. Dies wir mit der Indirekt und Spalten Funktion 

erreicht. 

=WENN(ISTFEHLER(FINDEN(Definition!$A2;INDIREKT("'Ausgabe 4'!B"&SPALTE(B$2))));" - 

"&TEXTVERKETTEN(ZEICHEN(10);WAHR;Definition!$A2;INDIREKT("'Ausgabe 4'!B"&SPALTE(B$2)));"") 

 Jetzt kann die Quellenliste und auch die Zelle mit der Eindeutig-Funktion nach rechts gezogen 

werden. 

 

4.2 Dynamisch machen des Verweises auf die Quelle 

Es muss jetzt in jeder Zeile auf ein bestimmt Spalte der Quelle verwiesen werden. Dies geht am 

besten mit der Funktion BEREICH.VERSCHIEBEN. Also aus dem Quellverweis: ='Quelle 4'!$A$24# wird: 

=BEREICH.VERSCHIEBEN('Quelle 4'!A$25#;0;ZEILE(A1)-1; ANZAHL2(BEREICH.VERSCHIEBEN('Quelle 

4'!A$25;0;ZEILE(A1)-1;20;1));1) 

Dabei wird mit Hilfe der Zeilenfunktion der Bereich der Quelle in jeder Zeile der Ausgabe um eine 

Spalte verschoben. 

Der blau markierte Bereich der Formel überprüft jeweils die Anzahl der noch zur Verfügung 

stehenden Länder und passt die Größe des Dropdowns entsprechend an.  

Jetzt kann die Quelle beliebig weit nach rechts und die Ausgabe beliebig weit nach unten erweitert 

werden. 

 

Also, die Mehrfachauswahl in der Datenüberprüfung geht auch ohne VBA       

q.e.d. 

 


